

# Using The 6522 to drive a Printer

Edward H. Carlson  
Okemos, MI

Low price compatible with good quality. If you are reaching the edge of your budget, the fifty dollars you can save by buying the parallel version of a printer may loom large. I wanted a printer for word processing and chose the Comprint 912P as suitable for rough draft printing. I was confident that the 6522 VIA on the CPU board of my Ohio Scientific C2-4P could handle the parallel interfacing. VIA stands for Versatile Interface Adaptor, and it can easily be configured to handle all the handshaking involved in the parallel transfer of data.

This article will describe how to wire the 6522 to the printer and will give a machine language program to drive it. The discussion is not at all restricted to OSI computers, nor even to the Comprint printer since the same principles apply to interfacing to other printers.

You may be interested in the features of the Comprint that appealed to me for word processing. It is fast, quiet and simple in design. The letter quality is high for a dot matrix printer as it has a 9x12 matrix. It is quiet because it is an electrostatic printer. This technology uses rolls of black paper which are coated with aluminum. The print head sparks holes through the aluminum to expose the black color below. The silvery paper is low in cost, thin and somewhat of a nuisance to handle. However, it Xeroxes very well. The 912 prints 3 lines a second of 80 characters each.

The Comprint has a variety of parallel options including the IEEE-488 convention and both wide and narrow strobe modes. I purchased the Comprint soon after it appeared on the market and made the modifications they suggested to operate with the Apple II Parallel Interface Card. (Since I also have an Apple, the same printer serves both computers.) The signal lines into the printer include seven parallel lines for the ASCII data and one line for DAV which is a narrow (one clock cycle is enough) strobe that tells the printer when valid data is on the 7 line bus. Signal lines from the Comprint include NDAC which goes low to acknowledge that the printer has accepted the character, and NRFD (not ready for data) which goes high when the printer's data buffer is full.

The 6522 VIA has two 8-bit ports, A and B, each with two control lines. The two ports are not identical and for no good reason I use the B port for the seven line ASCII bus. Since the eighth line is not needed for ASCII, I use it for the "busy" signal (NRFD). The B port control lines CB1 and CB2 are used for NDAC and DAV respectively.

The listing shows a subroutine, OUTCHR, that prints one character. Also included is a DRIVER that uses some subroutines in the OSI BASIC ROM's to read tape so its contents can be sent to the printer. Of course, this driver will need to be altered if your computer is not an OSI machine.

Implementing a 6522 can be a frustrating experience because of its many options. It has 16 registers of which we need 5. Three of the registers need be set only once, but we have plenty of time per character, and it is simpler to set these registers each time the subroutine is entered. Line 160

---

**...implementing a 6522 can be  
a frustrating experience because  
of its many options...**

---

enables the B port by setting bit 1 in the Auxiliary Control Register. In line 170, the Data Direction Register for B port is loaded such that lines 0 to 6 are output (for the ASCII character) and line 7 as input (for the DAV signal). Finally, the Peripheral Control Register must be tickled so that CB1 and CB2 know what is expected of them. This is done in line 210. Bits 7, 6, 5 are set to 100 so that CB2 will pulse low when the CPU writes to the VIA, (the strobe). Setting bit 4 tells the VIA to raise a flag when CB1 makes a low to high transition (the acknowledgement).

When the subroutine is entered, the accumulator A holds the character to be printed. It is saved by pushing it on the stack. Then the three registers mentioned above are configured. Next the VIA looks for the "busy" signal in lines 220 to 240. Upon finding a non-busy status, the character is pulled from the stack and sent to the B Output Register, and on to the printer. The last event is to detect the DAV acknowledgement. When it comes in on CB1, it sets a flag in the Interrupt Flag Register. Detecting this flag allows an exit from the loop of lines 300 to 330, and then exit from the subroutine.

There you have it. If you are interfacing to some other printer, the main thing to watch for is the polarity of the signal lines. Consult your 6522 data sheets for the code needed to reverse the polarity of the handshake signals. If by chance you have a Comprint 912P and have not configured it for Apple compatibility, I have written a program for that case too. An article describing it has been accepted for publication by BYTE. A copy of the program may be obtained by writing me at 3872 Raleigh Drive, Okemos, MI, 48864.

```

1 0000      ;      ***  TAPE TO COMPRINT 912P  ***
2 0000      ;
10 C000     *      =C000
20 C000 2007BF DRIVER JSR $BF07      GET CHAR. FROM TAPE PORT
25 C003 8D00D2      STA $D200      STORE CHAR. ON SCREEN
30 C006 200CC0      JSR OUTCHR      PRINT CHAR.
40 C009 4C00C0      JMP DRIVER
41 C00C      ;
42 C00C      ;      MY ADDRESSES, SEE FOOTNOTE
43 C00C      ;
44 C00C      VIA  =F700 ADDRESS OF 6522 IS $F7XX
46 C00C      AUX  =E0E  AUXILIARY CTRL REGISTER
48 C00C      BDD  =E08  B DATA DIRECTION REGISTER
50 C00C      BPORT =E00  OUTPUT REGISTER FOR I/O PORT B
52 C00C      PCTRL =E03  PERIPHERAL CONTROL REGISTER
54 C00C      IFLAG =E07  INTERRUPT FLAG REGISTER
60 C00C      ;
61 C00C      ;      STANDARD ADDRESSES
63 C00C      ;
64 C00C      ; VIA  PER YOUR MACHINE
66 C00C      ; AUX  =%1011
68 C00C      ; BDD  =%0010
70 C00C      ; BPORT =%0000
72 C00C      ; PCTRL =%1100
74 C00C      ; IFLAG =%1101
134 C00C     ;
140 C00C 4B    OUTCHR PHA      A CONTAINS CHARACTER
150 C00D A902   LDA #%00000010  ENABLE B PORT OF 6522
160 C00F 8D0EF7 STA VIA+AUX     AUX CTRL REGISTER
170 C012 A97F   LDA #%01111111  DATA DIRECTION
180 C014 8D08F7 STA VIA+BDD     B PORT DATA DIR REGISTER
190 C017 8D07F7 STA VIA+IFLAG    CLEAR INTERRUPT FLAGS
200 C01A A9B0   LDA #%10110000  PREPARE CB1 AND CB2
210 C01C 8D03F7 STA VIA+PCTRL    CB2 IS STROBE, PULSES LO
220 C01F AD00F7 BUSY LDA VIA+BPORT READ B PORT INPUT
230 C022 2980   AND #%10000000  BIT 7 IS NRFD OF COMPRINT
240 C024 30F9   BMI BUSY        BUSY IF BIT 7 IS HI
250 C026 68     PLA             LOAD CHAR. IN A
270 C027 8D00F7 STA VIA+BPORT    OUTPUT TO PRINTER
300 C02A AD07F7 ACK LDA VIA+IFLAG  LOOK FOR NDAC ON CB1
310 C02D 2910   AND #%00010000  MASK OUT DESIRED FLAG
320 C02F C910   CMP #%00010000  NDAC IS ACKNOWLEDGE
330 C031 D0F7   BNE ACK         IF NOT FOUND, LOOK AGAIN
340 C033 60     RTS
350 C034      ;
400 C034      ; COMPRINT PARALLEL I/O BOARD (PBC 1184 Rev C)
405 C034      ; HAS BEEN MODIFIED TO OPERATE WITH THE APPLE II
410 C034      ; PARALLEL PRINTER INTERFACE CARD
415 C034      ;
420 C034      ; THE 6522 HAS ADDRESS LINES 0,1 CONNECTED TO
422 C034      ; ADDRESSES 2,3 AND VICE VERSA

```